

Teoría de Lenguajes

Trabajo Práctico 1

Departamento de Computación
Facultad de Ciencias Exactas y Naturales
Universidad de Buenos Aires

Formateador de Código Jay

Integrante	LU	Correo electrónico
Aronson, Alex	443/08	<code>alexaronson@gmail.com</code>
Alvarez, Daniel		<code>salvarez@dc.uba.ar</code>
Nahabedian, Leandro	250/08	<code>leanahabedian@hotmail.com</code>
Ravasi, Nicolás	53/08	<code>nravasi@gmail.com</code>

Contents

1	Resolución	3
1.1	Transformación de la gramática del lexer	3
1.2	Transformación de la gramática de la sintaxis de Jay	4
1.3	Traducción dirigida por sintaxis	6
2	Referencias	7

1 Resolución

Analizando las gramaticas del lenguaje Jay dada por la catedra y el funcionamiento de las herramientas propuestas, pudimos observar que la gramatica no era $LL(1)$ debido a que tenía recursión a izquierda principalmente. Lo que decidimos hacer entonces es utilizar la herramienta *JavaCC* que si bien solo funciona para gramaticas ELL con un pequeño pasaje donde agregaremos los simbolos $*$, $+$ y $?$, podremos obtener facilmente la gramatica ELL deseada.

1.1 Transformación de la gramática del lexer

Dados los siguientes componentes léxicos de Jay:

```

InputElement  $\rightarrow$  WhiteSpace | Comment | Token
WhiteSpace  $\rightarrow$  space | \ t | \r | \n | \f | \r\n
Comment  $\rightarrow$  // any string ended by \r or \n or \r\n
Token  $\rightarrow$  Identifier | Keyword | Literal | Separator | Operator
Identifier  $\rightarrow$  Letter | IdentifierLetter | IdentifierDigit
Letter  $\rightarrow$  a | b | ... | z | A | B | ... | Z
Digit  $\rightarrow$  0 | 1 | 2 | ... | 9
Keyword  $\rightarrow$  boolean | else | if | int | main | void | while
Literal  $\rightarrow$  Boolean | Integer
Boolean  $\rightarrow$  true | false
Integer  $\rightarrow$  Digit | IntegerDigit
Separator  $\rightarrow$  ( ) | { } | ; | ,
Operator  $\rightarrow$  = | + | - | * | / | < | <= | > |
           | >= | == | != | && | || | !

```

Los reescribiremos de la siguiente manera:

```

InputElement  $\rightarrow$  WhiteSpace | Comment | Token
WhiteSpace  $\rightarrow$  space | \ t | \r | \n | \f | \r\n
Comment  $\rightarrow$  //.*$
Token  $\rightarrow$  Identifier | Keyword | Literal | Separator | Operator
Identifier  $\rightarrow$  Letter | IdentifierLetter | IdentifierDigit
Letter  $\rightarrow$  [a-zA-Z]
Digit  $\rightarrow$  [0-9]
Keyword  $\rightarrow$  boolean | else | if | int | main | void | while
Literal  $\rightarrow$  Boolean | Integer
Boolean  $\rightarrow$  true | false
Integer  $\rightarrow$  Digit | IntegerDigit
Separator  $\rightarrow$  ( ) | { } | ; | ,
Operator  $\rightarrow$  = | + | - | * | / | < | <= | > |
           | >= | == | != | && | || | !

```

Luego eliminamos las recursiones usando $*$ o $+$ y factorizamos utilizando $?$ entonces Integer va a ser:

```

Integer  $\rightarrow$  [0-9]+
Identifier  $\rightarrow$  [a-zA-Z][a-zA-Z0-9]*
Operator  $\rightarrow$  =(=)? | + | - | * | / | <(=)? |

```

| >(=)? | !(=)? | && | ||

Quedando la siguiente gramática

InputElement → *WhiteSpace* | *Comment* | *Token*
WhiteSpace → **space** | \ t | \r | \n | \f | \r\n
Comment → *//.*\$*
Token → *Identifier* | *Keyword* | *Literal* | *Separator* | *Operator*
Identifier → [a - zA - Z]([a - zA - Z0 - 9])*
Keyword → **boolean** | **else** | **if** | **int** | **main** | **void** | **while**
Literal → *Boolean* | *Integer*
Boolean → **true** | **false**
Integer → [0 - 9]⁺
Separator → (| { | } | ; | ,
Operator → =(=)? | + | - | * | / | <(=)? |
 | >(=)? | !(=)? | && | ||

1.2 Transformación de la gramática de la sintaxis de Jay

La idea ahora es realizar la misma acción para la gramática que define la sintaxis de Jay:

Program → **void main** () {*Declarations Statements*}
Declarations → λ | *Declarations Declaration*
Declaration → *Type Identifiers* ;
Type → **int** | **boolean**
Identifiers → *Identifier* | *Identifiers, Identifier*
Statements → λ | *Statements Statement*
Statement → ; | *Block* | *Assignment* | *IfStatement* | *WhileStatement*
Block → {*Statements*}
Assignment → *Identifier = Expression* ;
IfStatement → **if**(*Expression*) *Statement* | **if**(*Expression*) *Statement* **else** *Statement*
WhileStatement → **while**(*Expression*) *Statement*
Expression → *Conjunction* | *Expression* || *Conjunction*
Conjunction → *Relation* | *Conjunction && Relation*
Relation → *Addition* |
 | *Relation* < *Addition* | *Relation* <= *Addition* |
 | *Relation* > *Addition* | *Relation* >= *Addition* |
 | *Relation* == *Addition* | *Relation* != *Addition*
Addition → *Term* | *Addition* + *Term* | *Addition* - *Term*
Term → *Negation* | *Term* * *Negation* | *Term* / *Negation*
Negation → *Factor* | ! *Factor*
Factor → *Identifier* | *Literal* | (*Expression*)

Ahora vamos a eliminar las recursiones a izquierda como las de la producción *Declarations*, *Identifiers*, *Statements*, *Expression*, *Conjunction*, *Relation*, *Addition*, *Term*:

Declarations → (*Declaration*)*
Identifiers → *Identifier* (*Identifier*)*

Statements \rightarrow *Statement*^{*}
Expression \rightarrow *Conjunction* (*|| Conjunction*)^{*}
Conjunction \rightarrow *Relation* (&& *Relation*)^{*}
Relation \rightarrow *Addition* (*Comparison Addition*)^{*}
Comparison \rightarrow < | <= | > | >= | == | !=
Addition \rightarrow *Term* (*Sums Term*)^{*}
Sums \rightarrow + | -
Term \rightarrow *Negation* (*Mult Negation*)^{*}
Mult \rightarrow * | /

Note que para poder extender la producción *Relation*, *Addition* y *Term* tuvimos que agregar tres producciones auxiliares, *Comparison*, *Sums* y *Mult* respectivamente. Además como *Declarations* y *Statements* llaman a otra producción ninguna o muchas veces, vamos a eliminar estas dos producciones y colocar en su reemplazo el lado derecho del mismo. Pasemos ahora a colocar los ? que es otro caracter que extiende la gramática original y me va a reducir un poco mas lo que ya tengo. Agregaremos este símbolo en las siguientes producciones:

IfStatement \rightarrow **if**(*Expression*) *Statement* (**else** *Statement*)?
Negation \rightarrow (!)? *Factor*
Comparison \rightarrow <(=)? | >(=)? | == | !=

Veamos como quedan todos estos cambios al ponerlos todos juntos

Program \rightarrow **void main** () {*Declaration*^{*} *Statement*^{*}}
Declaration \rightarrow *Type* *Identifiers* ;
Type \rightarrow **int** | **boolean**
Identifiers \rightarrow *Identifier* (*Identifier*)^{*}
Statement \rightarrow ; | *Block* | *Assignment* | *IfStatement* | *WhileStatement*
Block \rightarrow {*Statement*^{*}}
Assignment \rightarrow *Identifier* = *Expression* ;
IfStatement \rightarrow **if**(*Expression*) *Statement* (**else** *Statement*)?
WhileStatement \rightarrow **while**(*Expression*) *Statement*
Expression \rightarrow *Conjunction* (*|| Conjunction*)^{*}
Conjunction \rightarrow *Relation* (&& *Relation*)^{*}
Relation \rightarrow *Addition* (*Comparison Addition*)^{*}
Comparison \rightarrow <(=)? | >(=)? | == | !=
Addition \rightarrow *Term* (*Sums Term*)^{*}
Sums \rightarrow + | -
Term \rightarrow *Negation* (*Mult Negation*)^{*}
Mult \rightarrow * | /
Negation \rightarrow (!)? *Factor*
Factor \rightarrow *Identifier* | *Literal* | (*Expression*)

Lo que nosotros elegimos como herramienta para hacer el trabajo práctico fue usar Javacc en el lenguaje Java. Al tener esta gramática definida pasamos a definir un archivo Jay.jj que se encargará de leer esta gramática y darnos el parser deseado.

1.3 Traducción dirigida por sintaxis

Ahora a la gramática vamos a tener que agregarle atributos para obtener luego de parsear todo el código Jay un código HTML.

Para mas detalle de esto, ver el código en Jay.jj.

2 Referencias

- 1) javacc.java.net